

# MES PREMIERS PAS EN

# BASIC



```
10 CLS
```

```
20 FOR K=1 TO
```

```
30 PRINT "LUNDI"
```

```
40 NEXT K
```

BOUCLE

ÉCRAN


Septembre 85.

~~Sp~~

ÉCOLE de DAMPIERRE sur LINOTTE  
70230 MONTBOZON

# MES PREMIERS PAS EN BASIC

Gaby Waters



Illustrations de Graham Round. Programmes écrits par R. Schifreen, N. Tatchell et P. Shreeve. Édition : Helen Davies pour l'édition anglaise, Patrick Baradeau pour l'édition française. Adaptation, assistance technique et conseil : European Media Business, 9 place des Ternes, 75017 Paris


**ÉCHOS-ÉLECTRONIQUE**

HACHETTE

# Sommaire


4	Qu'est-ce que le Basic ?	24	Utilisons IF... THEN
6	Les premières commandes Basic	26	Faisons des boucles
8	Faisons un programme	28	Programmes à boucles
10	Tapons le programme	30	Les nombres aléatoires
12	Premiers programmes	32	Jouons avec les nombres aléatoires
14	Initiation aux variables	33	Chassons les bogues
16	Utilisons les variables	34	Programmes et devinettes
18	La commande INPUT	44	Solutions
20	Jouons avec PRINT	48	Les commandes Basic
22	Faisons des comparaisons		

© Hachette, Paris, 1985 pour l'édition française.  
© Usborne, Londres, 1984.




# Avant-propos

Ce livre s'adresse aux débutants. Il va leur apprendre à écrire des programmes en Basic, le langage le plus utilisé sur les ordinateurs familiaux. C'est en fait très facile, et il suffit de maîtriser quelques commandes simples pour inventer plein de programmes amusants.



Les premières pages de ce livre t'expliquent comment faire un programme. Ensuite, on te présentera les principales instructions du Basic, illustrées par quelques programmes simples. Enfin, on te proposera des jeux et quelques devinettes.




Quelques commandes Basic changent suivant les types d'ordinateurs. Mais on te les indique clairement, et tu trouveras page 33 un tableau de conversions.




Page 44 à 47, tu trouveras les réponses aux devinettes, brièvement expliquées. Et si tes propres programmes ne marchent pas, regarde les conseils de la page 32.

# Qu'est-ce que le Basic ?

Si tu veux que ton ordinateur exécute tes ordres, il faut lui parler dans sa langue qui est le Basic. Ce langage utilisé sur la plupart des ordinateurs familiaux, ressemble beaucoup à l'anglais. Mais rassure-toi, il est nettement plus simple : il suffit de connaître quelques mots et quelques symboles.



Si tu lui donnes des ordres en français (ou même en anglais), il ne comprendra pas.




Les mots PRINT et INPUT sont des instructions Basic. Ils ordonnent à l'ordinateur de faire quelque chose.

Pour que l'ordinateur comprenne, il faut faire un programme, c'est-à-dire lui donner une liste d'instructions très précises. Regarde l'écran ci-dessus : c'est un exemple de programme.

On peut presque tout faire avec le Basic : dessiner, résoudre des problèmes, et même faire de la musique !

## Comment travaillent les ordinateurs




Quand tu donnes un programme à l'ordinateur, celui-ci commence par le traduire dans sa langue à lui,

le "langage machine" (c'est-à-dire en une série de signaux électriques). Toi aussi, tu pourrais écrire des

programmes en langage machine, mais c'est très difficile.

## Les autres langages



Il y a beaucoup d'autres langages que le Basic. Sur ce dessin, tu vois un exemple du langage Logo, qui utilise des commandes simples comme GAUCHE, DROITE, AVANCE


pour dessiner sur l'écran. Le Logo permet aussi de commander un petit robot, appelé la tortue, qui se déplace en dessinant sur le sol avec un crayon.

Ces instructions sont écrites en Fortran. C'est un langage très bien adapté aux problèmes mathématiques et scientifiques.

# Les premières commandes Basic

Sur la plupart des ordinateurs, il faut taper chaque commande, ou instruction Basic, lettre par lettre. Mais sur certains micro-ordinateurs familiaux il suffit d'appuyer sur une touche.

Pour commencer, voyons la commande PRINT, qui ordonne à l'ordinateur d'afficher quelque chose sur l'écran.



Ecrivons sur l'écran



Si tu veux que l'ordinateur affiche quelque chose, il faut lui en donner l'ordre en tapant PRINT (ce mot anglais signifie : imprime), suivi du mot choisi placé entre guillemets.




Si tu as fait une faute de frappe, corrige-la soigneusement, sinon l'ordinateur ne comprendra pas. Puis, appuie sur la touche marquée RETURN pour que l'ordinateur exécute l'ordre.



Quand tu appuies sur RETURN (ou sur ENTER), l'ordinateur écrit le mot sur l'écran\*. Familiarise-toi avec PRINT et RETURN en jouant à afficher d'autres mots.


## Affichons des messages



Regarde ci-dessous comment on fait pour effacer l'écran après chaque message.

Voilà quelques idées de messages. Affiche-les en utilisant à chaque fois PRINT, suivi d'un espace et du groupe de mots placés entre guillemets.

## Nettoyons l'écran



La commande qui permet d'effacer l'écran change d'un ordinateur à l'autre. Cherche dans le tableau de la page 33 celle qui correspond à ton appareil\*.



Quand tu as tapé cette commande, appuie sur RETURN : aussitôt, tout ce qui était sur l'écran disparaît, et le curseur revient en haut à gauche.

## Affichons des nombres

PRINT 1234

PRINT 999

PRINT 12

Pour afficher des nombres, on utilise aussi PRINT. Mais, cette fois, il n'y a pas besoin de guillemets. Affiche différents nombres en t'inspirant de ces exemples.

\* Chaque fois que tu verras CLS, utilise la commande qui efface l'écran de ton ordinateur.

# Faisons un programme


Imaginons que tu envoies un petit fantôme-ordinateur acheter une glace. Un ordre comme celui qui est écrit dans la bulle est bien trop difficile pour sa petite cervelle ! Il faut lui donner une série d'instructions très précises, qu'on appelle un programme.

Avant de l'écrire, réfléchis bien et fais la liste détaillée de tout ce que doit faire l'ordinateur.



## LISTE

1. Aller voir le marchand de glaces
2. Demander une grande glace à la fraise
3. Payer le marchand
4. Prendre la glace (sans la manger)
5. Faire demi-tour et revenir
6. Donner la glace à Mimi la Malice




Pense à ce que le petit fantôme doit faire à chaque étape, et dresse une liste d'instructions qu'il devra suivre pas à pas.

Concevons  
un programme

## PLAN DU PROGRAMME

1. Nettoyer l'écran
2. Demander le mot de passe
3. Attendre la réponse
4. Si la réponse est correcte  
afficher BRAVO
5. Si la réponse est  
mauvaise, afficher  
une blague.



Quand tu écris un programme, commence par faire le plan détaillé des ordres que tu dois donner à l'ordinateur.

```
10 CLS
20 PRINT "QUEL EST LE MOT DE PASSE?"
30 INPUT P$
40 IF P$="ANDOUILLE" THEN PRINT "BRAVO"
50 IF P$<>"ANDOUILLE" THEN PRINT "FAUX, MONSTRE DEBILE"
```

Tu trouveras l'explication de ces commandes Basic dans les pages suivantes.

Numérote toujours les lignes de 10 en 10 : ainsi tu pourras en ajouter plus tard sans aucune difficulté.

Voici ton plan transformé en programme : chaque ordre a été traduit en Basic et placé sur une ligne séparée. Chaque ligne doit commencer par un numéro qui indique à l'ordinateur l'ordre à suivre pour exécuter le programme.



# Tapons le programme

Quand tu tapes un programme au clavier, l'ordinateur stocke toutes les commandes dans sa mémoire. Pour les exécuter, il attend ton ordre.

Voici un programme qui affiche un poème « surréaliste ». Rentre-le dans ton ordinateur en suivant bien les conseils.

Pour afficher les vers du poème, le programme utilise l'instruction PRINT.

## Programme du poème

Rappelle-toi que cette commande est différente sur certains ordinateurs.

```
10 CLS  
20 PRINT "ROUGES SONT LES ROSES"  
30 PRINT "BLEUES LES VIOLETTES"  
40 PRINT "MAIS AVEC MA TETE"  
50 PRINT "LA VIE N'EST PAS ROSE"
```

Ne confonds pas le chiffre 0 (toujours barré en informatique) avec la lettre O : pour l'ordinateur, ce n'est pas du tout pareil !

## Tapons le programme



N'oublie pas de taper le numéro placé au début de chaque ligne.



Sur ton ordinateur, la touche DELETE peut porter un autre nom. Regarde dans ton manuel d'utilisation.



Quand tu lances le programme, voici ce que tu dois obtenir.

Tape le programme très attentivement. À la fin de chaque ligne, vérifie qu'il n'y a pas de faute, puis appuie sur ENTER pour que l'ordinateur stocke cette ligne en mémoire.

Si tu fais une faute de frappe, efface-la en appuyant sur la touche DELETE ou DEL.

Quand tu as rentré toutes les lignes, tape RUN et appuie sur RETURN : tu ordonnes ainsi à l'ordinateur d'exécuter le programme.

## Les erreurs



Voici un exemple de message d'erreur : erreur de syntaxe à la ligne 20.

Si tu fais une faute, même une simple faute de frappe (oubli d'une lettre, de guillemets) le programme ne marchera pas. On appelle ces fautes des bogues (ou "bugs"); nous verrons page 33 comment les repérer.

## Allons plus loin...



Tu peux rentrer les lignes du programme dans n'importe quel ordre ; l'ordinateur les remettra automatiquement dans l'ordre croissant. Vérifie-le avec le programme

## Invitation à mon anniversaire

```
30 PRINT "TU EST INVITE A"  
70 PRINT "UN JOLI CADEAU"  
60 PRINT "APPORTE-MOI"  
20 PRINT "CHER GROS VILAIN"  
50 PRINT "SAMEDI PROCHAIN"  
10 CLS  
80 PRINT "AMITIES. MIMI"  
40 PRINT "MON ANNIVERSAIRE"
```

ci-dessus.

Tu peux aussi demander à l'ordinateur d'afficher (on dit "lister") le programme : fais LIST (suivi de RETURN), et toutes les lignes seront affichées dans le bon ordre.

## Pour effacer un programme



Quand tu tapes NEW, certains ordinateurs effacent aussi l'écran.

Quand tu tapes NEW (toujours suivi de RETURN), le programme est effacé de la mémoire de l'ordinateur.

Pour modifier une ligne, retape son numéro suivi de la nouvelle version.

20

70 PRINT "UN GATEAU AU CHOCOLAT"

75 PRINT "(J'ADORE LE CHOCOLAT)"


Pour supprimer une ligne, tape simplement son numéro et appuie sur RETURN.

Pour ajouter des lignes, il suffit de leur donner des numéros nouveaux : l'ordinateur les rangera automatiquement à la bonne place.

Un programme se modifie aisément : rien de plus facile que de corriger, d'ajouter ou de supprimer des lignes. Essaie d'apporter ces modifications au programme d'anniversaire.

# Premiers programmes

Avec l'instruction PRINT, tu peux aussi utiliser des lettres ou des symboles pour faire des dessins schématiques. Tape ce programme qui dessine une fusée.



## Programme de la fusée


```
10 CLS
20 PRINT "   □□□"
30 PRINT "  □□!□!"
40 PRINT " □!!□!!"
50 PRINT "□!!□!!"
60 PRINT "□!!□!!"
70 PRINT "□!!□!!"
80 PRINT "XXX□XXX"
```

Chaque PRINT affiche une ligne de symboles et d'espaces. N'oublie pas les guillemets au début et à la fin de chaque ligne.


Commence toujours par une instruction CLS pour nettoyer l'écran.

Ce petit carré t'indique que tu dois taper un espace. À chaque □, appuie une fois sur la barre d'espace.

## Programme du visage



Commence toujours par mettre l'instruction qui efface l'écran.




Compare ton programme à la solution de la page 44.

Regarde bien ce visage et essaie d'écrire le programme correspondant.

Chaque ligne du dessin sera affichée grâce à un PRINT. Pour trouver la disposition exacte des symboles et des

espaces, reproduis le dessin sur du papier quadrillé : chaque carré correspondra à un symbole ou à un espace.

## Un peu de calcul



Fais faire quelques opérations à l'ordinateur.

Si tu tapes PRINT suivi d'une opération, l'ordinateur va faire le calcul et afficher le résultat. Essaie avec ces exemples. Remarque que ces lignes n'ont pas de numéro : l'ordinateur exécute donc immédiatement chaque commande.

Pour multiplier, on utilise le signe \* ; pour diviser, le signe /. Pour additionner ou soustraire, on emploie les signes habituels.

## Les heures de l'année

10 CLS

20 PRINT "IL Y A"

30 PRINT 365\*24

40 PRINT "HEURES DANS"

50 PRINT "UNE ANNEE"

Essaie sur ton ordinateur ce programme qui calcule le nombre d'heures contenues dans une année. La ligne 30 dit à l'ordinateur de multiplier le nombre de jours contenus dans une année par le nombre d'heures contenues dans une journée et d'afficher le résultat.

Une journée est divisée en 24 heures.

Dans une année, il y a 365 jours.


Voilà ce que tu dois obtenir quand tu fais RUN (n'oublie pas RETURN!).



# Initiation aux variables

Quand on fait un programme, on donne généralement à l'ordinateur les informations nécessaires pour l'exécuter. Celui-ci va mettre ces renseignements dans des sortes de boîtes, qu'on appelle des variables, qu'il va ranger dans sa mémoire.

Voyons comment on utilise l'une des commandes qui permettent de donner des informations à l'ordinateur : LET.



Une mémoire ressemble beaucoup aux cases vides de ce dessin. Quand tu donnes une information à l'ordinateur, tu dois lui donner aussi une

étiquette. L'ordinateur range l'information dans une boîte sur laquelle il "colle" l'étiquette : voilà ce qu'on appelle une variable.

## L'emploi de LET




Voici l'étiquette (ou le nom) de la variable.

LET N=999


Et voilà le nombre stocké dans cette variable.

Cette commande LET dit à l'ordinateur de stocker la valeur 999 dans la variable appelée N.

DANS L'ORDINATEUR



## Devinettes : les nombres



Essaie d'écrire les trois commandes qui donnent aux variables G, Z, C, les valeurs ci-dessus. Si tu ne trouves pas, regarde la solution en page 44.

## Les chaînes de caractères



En Basic, tout ce qui est mot, lettre, signe ou symbole est appelé chaîne de caractères. Si tu veux stocker une chaîne de caractères, tu dois la mettre entre guillemets. Et le nom de la variable doit être suivi du signe \$ (dollar).



### Utilisons PRINT avec des variables

```
10 CLS  
20 LET N=99  
30 LET B$="BANANES"  
40 PRINT N  
50 PRINT B$
```

Ici, la commande PRINT ordonne à l'ordinateur d'afficher le contenu des variables N et B\$.

Si tu veux que l'ordinateur affiche l'information stockée dans une variable, tape PRINT suivi du nom de cette variable. Rentre ce programme et regarde comment il fonctionne.



### Devinette : la phrase

```
10 CLS  
20 LET A$="C'EST"  
30 LET B$="TROP"  
40 LET C$="DUR"
```




Ajoute trois instructions PRINT à ce programme pour que ton ordinateur affiche les informations contenues dans A\$, B\$ et C\$. (Solution page 44)

# Utilisons les variables

Voici quelques conseils concernant les noms des variables et quelques exemples d'utilisation.

Un bon conseil : choisis des noms qui évoquent ce qui est contenu dans les variables.

## Les noms des variables



Généralement, les noms (ou les étiquettes) des variables sont des lettres. Mais, sur certains ordinateurs, on peut utiliser une lettre suivie d'un

chiffre, ou même un mot. Regarde dans le manuel de ton ordinateur quels noms de variables il accepte.

## Les noms interdits



Attention : ne prends pas comme noms de variables des commandes Basic ! L'ordinateur ne comprendrait plus, et tu aurais droit à un message d'erreur !

## Juxtaposons les variables et les chaînes de caractères

1

```
10 CLS
20 LET N=7
30 PRINT N;" PETITS COCHONS"
40 PRINT "MANGENT ";N;" JAMBONNEAUX"
```

Fais attention au point-virgule.

Si tu oublies ces espaces, le contenu de la variable N sera collé aux mots placés avant et après lui.

2

```
7 PETITS COCHONS
MANGENT 7 JAMBONNEAUX
```

Voici ce que tu dois obtenir si tu ne fais pas de faute.

Pour juxtaposer un mot et le contenu d'une variable, utilise PRINT suivi du mot placé entre guillemets, d'un point-

virgule et du nom de la variable. Le point-virgule est une commande Basic très pratique : il ordonne à

l'ordinateur de continuer sur la même ligne au lieu de passer à la ligne suivante.

## Un peu de calcul

1

```
10 CLS
20 LET A=2
30 LET B=5
40 PRINT A+B
50 PRINT A*B
```

L'ordinateur affiche le résultat des opérations.




2

```
35 PRINT A;"+";B;"=";
45 PRINT A;"*";B;"=";
```

2+5=7  
2x5=10

Si tu mets un point-virgule à la fin d'une ligne, l'ordinateur continuera à écrire sur la même ligne.



On utilise aussi les variables pour faire des calculs. Dans ce programme, l'ordinateur additionne, puis multiplie les nombres contenus dans les variables A et B, et affiche les résultats.

Bien souvent, on se demande à quoi correspond un résultat affiché tout seul, sans explication. Ajoute donc ces deux lignes PRINT pour que l'ordinateur explique ce qu'il fait.

## Programme des gâteaux

```
10 CLS
20 LET G=12
30 LET M=4
40 LET S=G/M
50 PRINT "IL Y A ";S;" GATEAUX"
60 PRINT "POUR CHAQUE MONSTRE"
```

Lance le programme pour voir combien de gâteaux mangera chaque monstre.




Quatre monstres gourmands veulent se partager équitablement 12 gâteaux. Ce programme va les aider à trouver la

solution. La variable G contient le nombre de gâteaux, M le nombre de monstres. La variable S, c'est la solution.

# La commande INPUT

Lorsque tu utilises LET, le contenu de la variable est écrit dans le programme : chaque fois que tu l'utilises, il reprend les mêmes informations et donne donc les mêmes résultats.

La commande INPUT, elle, permet de donner des informations à l'ordinateur pendant l'exécution d'un programme. Tu peux donc rentrer des informations différentes à chaque fois.



Après INPUT, on met toujours un nom de variable. Quand l'ordinateur rencontre INPUT, il "colle" ce nom sur une case vide de sa mémoire, et il attend que tu tapes quelque chose.



Pour te montrer qu'il attend une information, l'ordinateur affiche un point d'interrogation. Ta réponse, il va la mettre dans la case réservée, c'est-à-dire dans la variable.


## Exemple de programme

Cette ligne PRINT affiche la question posée.

```
10 CLS
20 PRINT "QUEL EST TON NOM?"
30 INPUT N$
40 PRINT "QUEL AGE AS-TU?"
50 INPUT A
60 CLS
70 PRINT N$
80 PRINT "A ";A;" ANS"
```

Cette ligne dit à l'ordinateur d'attendre un mot.


Cette ligne lui dit d'attendre un nombre.



Ce programme utilise deux INPUT pour te demander ton nom et ton âge, puis il les affiche sur l'écran. Chaque fois que tu lanceras ce

programme, il te reposera les questions : tu peux donc donner n'importe quelle réponse, même la plus débile.

## L'ordinateur "calculette"



Tu dois stocker les nombres dans deux variables appelées A et B.

Cette ligne dit à l'ordinateur d'additionner A et B et de stocker la réponse en C.

```
10 CLS
20 PRINT "DONNE UN NOMBRE"
30
40 PRINT "DONNE UN AUTRE NOMBRE"
50
60 LET C=A+B
70 PRINT A;"+";B;"=";C
```

Ce programme fait travailler ton ordinateur comme une calculette. Quand tu as tapé les deux nombres, l'ordinateur les additionne et

affiche le résultat. Complète les lignes 30 et 50 avec des INPUT. Si tu n'y arrives pas, regarde la solution page 44.

## Devinette : le message

```
10 CLS
20 PRINT "TAPE UN MESSAGE"
30 INPUT M$
40 PRINT "NOM DU DESTINATAIRE"
50 INPUT D$
60 PRINT "NOM DE L'EXPEDITEUR"
70 INPUT E$
80 CLS
90 PRINT
100 PRINT M$
110 PRINT
```

Voici les instructions que tu dois compléter. Vérifie ta réponse page 44.

MIMI LA MALICE  
RENDS-MOI VITE  
MON BEAU BALLON  
TOUT NEUF  
GROS VILAIN

Encore un programme à compléter ! L'ordinateur demande un message, puis à qui il est adressé, et enfin, qui l'envoie. L'ordinateur stocke

ces informations dans les variables M\$, D\$ et E\$. Puis il efface l'écran et affiche le message. Peux-tu compléter les

commandes PRINT des lignes 90 et 110 pour que l'ordinateur affiche aussi, à la bonne place, les noms stockés en D\$ et E\$ ?

# Jouons avec PRINT

Et maintenant, voyons comment afficher quelque chose à n'importe quel endroit de l'écran.

## PRINT TAB



La commande PRINT TAB ordonne à l'ordinateur de laisser des espaces avant un

mot ou un nombre. Pour lui indiquer combien d'espaces il doit laisser, tu mets le nombre



voulu entre parenthèses. Lance ce programme pour voir comment il fonctionne.

## Petit programme avec PRINT TAB



```
10 CLS
20 PRINT "COMBIEN D'ESPACES"
30 PRINT "VEUX-TU LAISSER?"
40 INPUT N
50 PRINT TAB(N);"BONJOUR"
```

Ce programme te demande combien d'espaces tu veux laisser avant le mot BONJOUR. L'ordinateur stocke le nombre dans la variable N, qu'il utilise

ensuite avec l'instruction PRINT TAB. Essaie ce programme à plusieurs reprises en changeant de nombre à chaque fois.

## Pour sauter des lignes

```
10 CLS
20 PRINT "BONJOUR"
30 PRINT
40 PRINT
50 PRINT
60 PRINT "ADIEU"
```

Pour laisser une ligne vide, il suffit de dire à l'ordinateur d'afficher... rien du tout : on utilise donc PRINT tout seul, comme dans ce programme.

## Programme pour un anniversaire

Remarque les deux-points.

Tu peux mettre plusieurs instructions sur une ligne, à condition de les séparer par des deux-points.

```
10 CLS
20 PRINT:PRINT:PRINT
30 PRINT:PRINT:PRINT
40 PRINT TAB( );"BON ANNIVERSAIRE"
50 PRINT TAB( );"MIMI"
60 PRINT TAB( );"*****"
```


Ce programme évoque une carte d'anniversaire. Rentre-le en mettant des nombres dans les parenthèses placées après PRINT TAB. Modifie-les jusqu'à ce que tu parviennes à afficher le message au milieu de l'écran, comme sur le dessin ci-contre.



## Faire des colonnes

```
10 CLS
20 PRINT:PRINT:PRINT
30 PRINT "JOUR", "MENU"
40 PRINT
50 PRINT "LUNDI", "CHOCOLAT"
60 PRINT "MARDI", "HAMBURGER"
70 PRINT "MERCREDI", "GLACE"
80 PRINT "JEUDI", "FRITES"
90 PRINT "VENDREDI", "POISSON"
100 PRINT "SAMEDI", "CHIPS"
110 PRINT "DIMANCHE", "POPCORNS"
```

Essaie ce programme. Voici ce que tu dois voir sur ton écran.



Pour faire des colonnes, on met une virgule après un mot, un nombre ou une variable :

l'ordinateur affichera ce qui suit au début de la colonne suivante.

# Faisons des comparaisons

Un ordinateur sait aussi comparer deux informations. On peut donc lui dire de faire telle ou telle chose en fonction du résultat. Pour cela, on utilise l'instruction IF... THEN (en français : SI telle condition est remplie... ALORS fais ceci).

```
10 CLS
20 PRINT "DONNE DEUX NOMBRES"
30 INPUT A
40 INPUT Z
50 IF A=Z THEN PRINT "EGAUX"
60 IF A<>Z THEN PRINT "DIFFERENTS"
```

Tape un nombre et appuie sur RETURN, puis recommence pour le second.



Ce symbole signifie : regarde si ces nombres sont identiques.



Ce symbole signifie : regarde si ces nombres sont différents.

Dans ce programme, la commande IF... THEN ordonne à l'ordinateur de comparer les nombres stockés dans les variables A et Z. Et le

résultat de la comparaison est affiché sur l'écran.

## D'autres façons de comparer

L'ordinateur sait aussi comparer deux nombres pour dire si l'un est plus grand, ou plus petit que l'autre. Ces dessins te montrent les symboles que l'on utilise.



## Comparer des mots

Tu peux aussi utiliser IF... THEN pour comparer des mots. Dans ce programme, l'ordinateur compare les mots contenus dans M\$ et N\$ pour voir si ce sont les mêmes.

À la place, tu peux mettre ton nom.

```
10 CLS
20 LET M$="MIMI LA MALICE"
30 PRINT "QUEL EST TON NOM?"
40 INPUT N$
50 IF N$<>M$ THEN PRINT "JE HAIS LES MONSTRES APPELES ";N$
60 IF N$=M$ THEN PRINT "BONJOUR ";M$;" CONTENT DE TE VOIR"
```

Comme ce message est long, certains ordinateurs ne l'afficheront pas sur une ligne, mais sur deux.

QUEL EST TON NOM  
? ROSETTE  
JE DETESTE LES  
MONSTRES APPELES  
ROSETTE

## Les âges

```
10 CLS
20 LET A=11
30 PRINT "J'AI ";A;" ANS"
40 PRINT "QUEL AGE AS-TU?"
50 INPUT B
60 IF A=B THEN PRINT "NOUS AVONS LE MEME AGE!"
70 IF A>B THEN PRINT "JE SUIS PLUS VIEUX QUE TOI"
80 IF A<B THEN PRINT "JE SUIS PLUS JEUNE QUE TOI"
```

À la place de 11, mets ton âge.

J'AI 11 ANS  
QUEL AGE AS-TU?  
? 10  
JE SUIS PLUS  
VIEUX QUE TOI

Quand tu rentres le programme, tape ton âge à la ligne 20. Puis fais jouer un camarade.

L'ordinateur te dira si ton camarade a le même âge, est plus jeune ou plus vieux que toi.

# Utilisons IF... THEN

Après une instruction IF... THEN, tu peux placer des ordres variés. Par exemple, tu peux dire à l'ordinateur d'aller à une autre ligne de programme ou de s'arrêter.

## IF... THEN GOTO

```
10 CLS
20 LET C=17
30 PRINT "DEVINE COMBIEN DE GATEAUX"
40 PRINT "A MANGE MIMI LA MALICE"
50 INPUT G
60 IF G<>C THEN GOTO 50
70 PRINT "T'ES UN CHEF!"
```

Si tu t'es trompé, ce GOTO renvoie l'ordinateur à la ligne 50.

Voilà ce qu'affiche l'ordinateur si ta réponse est bonne.

Dans ce programme, tu dois deviner le nombre de gâteaux mangés par cette gourmande de Mimi. Si tu te trompes,

GOTO (en français, VA EN...) renvoie l'ordinateur à la ligne 50 pour te donner une

nouvelle chance ; et cela continuera jusqu'à ce que tu trouves la bonne réponse.

## Améliorons le programme

Pour que ce programme soit plus clair, il faudrait que l'ordinateur affiche "FAUX" tant que tu n'as pas donné la bonne réponse. Pour cela, il suffit d'ajouter une commande PRINT après THEN.

Tape cette nouvelle version de la ligne 60. Qu'est-ce que cela change au programme ?

```
60 IF G<>C THEN PRINT "FAUX":GOTO 50
```

Attention : PRINT et GOTO doivent être sur la même ligne que IF... THEN. N'oublie pas de mettre le signe deux-points.

## IF... THEN STOP

```
1
10 CLS
20 LET P$="PETIT POIS"
30 PRINT "QUEL EST LE MOT DE PASSE?"
40 INPUT A$
50 IF A$<>P$ THEN STOP
60 PRINT "EXACT"
70 PRINT "BIENVENU, CHER AMI"
```

L'ordinateur n'affichera ces lignes que si tu donnes le bon mot de passe.

Voici le mot de passe. On le met dans la variable P\$.

Ce n'est pas le mot de passe!

2  
QUEL EST LE MOT DE PASSE?  
?HARICOT

STOP AT LINE 50


Ce message indique que le programme a été arrêté.

Le programme utilise une nouvelle instruction, STOP, placée après IF... THEN. Si tu ne tapes pas le bon mot de

pas, l'ordinateur s'arrête en affichant un message qui indique à quelle ligne du

programme il est stoppé. Ce message varie suivant les ordinateurs.

## La machine à calculer



```
10 CLS
20 PRINT "DONNE-MOI DEUX NOMBRES"
30 INPUT X:INPUT Y
40 PRINT "DOIS-JE ADDITIONNER"
50 PRINT "OU MULTIPLIER?"
60 PRINT "(TAPE A OU M)"
70 INPUT R$
80 IF R$="A" THEN LET Z=X+Y
90 IF R$="M" THEN LET Z=X*Y
100 PRINT "LA RÉPONSE EST ";Z
```

Rappelle-toi que tu peux placer deux instructions sur la même ligne à condition de les séparer par le signe deux-points.

La réponse est stockée en Z.

Essaie ce programme. L'ordinateur te demande d'abord deux nombres, puis si

tu veux les additionner ou les multiplier. Les instructions IF... THEN LET des lignes 80 et

90 lui permettent d'effectuer l'opération et de stocker le résultat dans la variable Z.

# Faisons des boucles

Une boucle, c'est une série d'instructions qui forcent l'ordinateur à répéter plusieurs fois la même chose. Pour faire une boucle, on emploie les commandes FOR... TO (POUR... A) et NEXT (SUIVANT).

Voici un exemple de boucle. Ce petit programme affiche cinq fois le même message.

Tu peux utiliser une boucle pour répéter un message comme celui-ci.

BONJOUR GROS VILAIN  
BONJOUR GROS VILAIN  
BONJOUR GROS VILAIN  
BONJOUR GROS VILAIN  
BONJOUR GROS VILAIN

## La boucle de Gros Vilain

```
10 CLS  
20 FOR K=1 TO 5  
30 PRINT "BONJOUR, GROS VILAIN"  
40 NEXT K
```

Cette commande PRINT est dans la boucle.

La boucle commence ici.

La boucle se termine ici.

Dans ce programme, la boucle commence à la ligne 20. Les mots FOR... TO indiquent à

l'ordinateur combien de fois il doit répéter l'instruction PRINT. À la ligne 40, le mot

NEXT lui ordonne de revenir au début de la boucle, c'est-à-dire à la ligne 20.

## Fonctionnement d'une boucle

Pour comprendre comment ça marche, pense à une course à pied : le coureur doit faire plusieurs fois le tour de la piste. Eh bien, chaque fois que l'ordinateur exécute l'instruction contenue dans la boucle, il fait "un tour de piste".

J, c'est la variable de cette boucle.

```
10 CLS
20 FOR J=1 TO 6
30 PRINT "BONJOUR"
40 NEXT J
```

Dans ce programme, la variable J sert à compter les tours. Au premier tour, J a la

valeur 1. Ensuite, chaque fois que l'ordinateur recommence la boucle, ce nombre augmente

d'une unité. Quand il arrive à 6, la boucle est finie.

## Une boucle à la loupe

```
10 CLS
20 FOR T=1 TO 20
30 PRINT "TOUR ";T
40 NEXT T
```

Avec ce programme, l'ordinateur affiche chaque passage dans la boucle, pour que tu le voies faire ses tours.

## Devinette : la boucle

Tape les chiffres manquants.


```
10 CLS
20 FOR J= TO
30 PRINT
40 NEXT J
```

Ici, tape ton nom, entre parenthèses.

Complète ce programme de telle sorte que l'ordinateur affiche dix fois ton nom. Quels nombres dois-tu mettre à la ligne 20 ? Écris ton nom entre guillemets après le PRINT de la ligne 30.

# Programmes à boucles

Voici quelques exemples d'utilisation de boucles dans des programmes.



```
10 CLS
20 PRINT "JUSQU'OU"
30 PRINT "DOIS-JE COMPTER?"
40 INPUT N
50 FOR K=1 TO N
60 PRINT K
70 NEXT K
```


Les boucles sont très pratiques pour faire compter les ordinateurs. Essaie ce programme : l'ordinateur te

demande jusqu'à quel nombre tu veux qu'il compte, puis il affiche les nombres les uns en dessous des autres.

## Différentes façons de compter

Cette ligne ordonne à l'ordinateur de compter de 5 en 5.

```
10 CLS
20 FOR L=0 TO 30 STEP 5
30 PRINT L
40 NEXT L
```



Amuse-toi à changer le nombre placé après STEP.

## Le compte à rebours

```
10 CLS
20 PRINT "COMPTE A REBOURS"
30 FOR J=10 TO 1
STEP -1
40 PRINT J
50 NEXT J
60 PRINT "FEU!"
```

Si tu en as assez de voir ton ordinateur compter normalement, tu peux lui dire de progresser par bonds successifs, comme quand tu

montes des marches 2 par 2. Par exemple, dans ce programme, pour lui ordonner de compter de 5 en 5, on a ajouté STEP (PAS) 5.

STEP suivi d'un nombre négatif fait compter l'ordinateur à l'envers. Lance ce programme pour voir comment il fonctionne.

## Pour faire attendre l'ordinateur

**1**

```
10 CLS
20 PRINT "BONJOUR"
30 FOR K=1 TO 1000
40 NEXT K
50 PRINT "RE-BONJOUR!"
```

Voici la boucle d'attente : elle force l'ordinateur à compter de 1 à 1000 avant de passer à la suite du programme.

Dans ce programme, il n'y a pas d'instruction entre FOR... TO et NEXT : l'ordinateur compte simplement de 1 au nombre indiqué après TO, mais il ne se passe rien sur l'écran. Tu as donc l'impression (fausse !) qu'il fait une pause : on appelle cela une boucle de temporisation.

## Mets une variable dans ta boucle

**1**

Chaque fois que l'ordinateur fait un tour, il multiplie 9 par le nombre contenu dans K, et il affiche le résultat.

```
10 CLS
20 FOR K=1 TO 12
30 LET A=K*9
40 PRINT K;"×9=";A
50 NEXT K
```

Ce programme affiche la table des 9. La variable de boucle, K, sert à la fois à compter les tours de la boucle et à calculer les résultats.

**2**

Un nombre plus petit raccourcit la pause.

```
30 FOR K=1 TO 500
30 FOR K=1 TO 2000
```

Un nombre plus grand allonge la pause.

Pour modifier la durée de la pause, il suffit de changer le nombre placé après TO. Amuse-toi à raccourcir, puis à allonger la temporisation du programme précédent.

**2**

Voici ce qui doit apparaître sur l'écran quand tu lances le programme.

```
10 CLS
20 FOR J=1 TO 10
30 PRINT TAB(J);"BONJOUR"
40 NEXT J
```

Dans ce programme, la variable de boucle, J, est aussi utilisée avec l'instruction PRINT TAB : à chaque tour de boucle, l'ordinateur laisse un espace de plus avant BONJOUR.

# Les nombres aléatoires

Quand on joue aux dés, il est très difficile de prédire les chiffres qui vont sortir. Ces nombres, tirés au hasard, sont les nombres aléatoires.

Voyons comment demander à l'ordinateur de tirer des nombres au hasard. L'instruction utilisée est assez compliquée, mais elle est bien pratique pour les programmes de jeu.



Pour faire tirer à l'ordinateur un nombre aléatoire, on utilise la commande INT suivie de RND(1) (ou RND tout seul, ou RND(0), sur d'autres ordinateurs) et de chiffres.

## Afficher des nombres aléatoires

```
10 CLS
20 PRINT "VOICI UN NOMBRE"
30 PRINT "COMPRIS ENTRE 1 ET 20"
40 PRINT INT(RND(1)*20+1)
```

On va tirer au hasard un nombre compris entre 1 et 20.

N'oublie pas d'adapter cette formule à ton ordinateur, si c'est nécessaire.

VOICI UN NOMBRE  
COMPRIS ENTRE  
1 ET 20  
17

Il peut arriver que l'ordinateur tire deux fois le même nombre.

Voici un programme simple de tirage au sort. À la ligne 40, l'ordinateur choisit un nombre

aléatoire et l'affiche sur l'écran. Essaie ce programme plusieurs fois de suite :

l'ordinateur affichera à chaque fois des nombres différents.

2

Ici, tu mets le nombre maximal que peut choisir l'ordinateur.

$\text{INT}(\text{RND}(1)*6+1)$

Ici, tu mets le nombre le plus petit.

3

DANS L'ORDINATEUR

Pour compléter l'instruction RND, il faut ajouter deux chiffres :  
le premier indique le nombre de choix que peut faire l'ordinateur, le second donne le plus petit nombre qu'il peut

choisir. Par exemple, quand tu joues avec un dé, tu peux tirer six chiffres différents, le plus petit étant 1. Regarde ci-dessus comment l'ordinateur joue avec le dé.

Si tu donnes à l'ordinateur l'instruction dessinée dans le cadre 2, il choisira "les yeux bandés" un nombre compris entre 1 et 6.

### Programme de loterie



Adapte cette formule à ton ordinateur.

Remplace ces espaces par les nombres qui conviennent.

```
10 CLS
20 LET W=INT(RND(1)*  + )
30 PRINT "SALUT LES MONSTRES"
40 PRINT "L'HEUREUX GAGNANT"
50 PRINT "EST LE NUMERO ";W
```

Nos gentils monstres jouent à la loterie. Ils ont 30 billets numérotés de 1 à 30, et ils veulent que l'ordinateur désigne le vainqueur. Le programme qui va faire le tirage au sort est prêt, mais il est

incomplet : il manque deux nombres à la ligne 20. Que faut-il mettre pour que l'ordinateur tire un nombre compris entre 1 et 30 ?

# Jouons avec les nombres aléatoires

Essaie ces deux programmes qui utilisent la commande RND (abréviation de RANDOM, qui signifie hasard en Anglais).

## Devinette

```
10 CLS
20 PRINT:PRINT:PRINT
30 LET R=INT(RND(1)*20+1)
40 PRINT TAB(R);"X"
50 PRINT "QUELLE EST LA VALEUR DE TAB?"
60 INPUT G
70 IF G=R THEN PRINT "BRAVO!"
80 IF G<>R THEN PRINT "NON, C'ETAIT ";R
```

Attention : sur ton ordinateur, tu dois peut-être remplacer RND(1) par RND.



```

X
QUELLE EST LA VALEUR DE TAB?
?7
NON, C'ETAIT 10
```

Le PRINT de la ligne 40 commande à l'ordinateur d'afficher un X sur l'écran ; mais le nombre correspondant à TAB est choisi au hasard. Quand tu joues, l'ordinateur te demande de deviner ce nombre d'après la position du X sur l'écran. Si tu te trompes, il te donne la bonne réponse.

## Calcul mental

```
COMBIEN FONT 7+9?
?16
EXACT
```

```
10 CLS
20 LET X=INT(RND(1)*12+1)
30 LET Y=INT(RND(1)*12+1)
40 LET Z=X+Y
50 PRINT "COMBIEN FONT ";X;"+";Y;"?"
60 INPUT A
70 IF Z=A THEN PRINT "EXACT"
80 IF Z<>A THEN PRINT "NON, CELA FAIT ";Z
```

Es-tu bon en calcul mental ? Ce petit programme va te le dire. L'ordinateur choisit deux

nombres aléatoires compris entre 1 et 12 ; puis il te demande de les additionner et

de taper le résultat. Enfin, il te donne la réponse si tu t'es trompé.

# Chassons les bogues

Rappelle-toi que les bogues, ce sont les erreurs qui se sont glissées dans les

programmes (et il y en a très souvent !).  
Voici comment on les corrige.



Pour t'aider à trouver les bogues, ton ordinateur affiche des messages différents selon les types d'erreurs que tu as commis. Cherche leur signification dans ton manuel.



Pour trouver l'erreur, tape LIST et appuie sur RETURN : toutes les lignes du programme s'affichent automatiquement sur l'écran. Relis très attentivement chaque ligne.



Quand tu as trouvé l'erreur, retape toute la ligne, numéro compris, sans faire de faute cette fois ! (Il y a d'autres façons de corriger que tu apprendras plus tard.)

## Tableau de conversions

Ce tableau précise les commandes utilisées par quelques ordinateurs pour nettoyer l'écran et choisir des nombres aléatoires.

Complète cette commande avec des nombres, comme tu l'as appris page 31.

	Efface l'écran	Tire un nombre aléatoire
<b>T07</b>	CLS	INT(RND* + )
<b>VIC 20 et Commodore 64</b>	PRINT CHR\$(147)	INT(RND(1)* + )
<b>Apple</b>	HOME	INT(RND(1)* + )
<b>Spectrum</b>	CLS	INT(RND* + )

# Programmes et devinettes

Et maintenant, jouons avec les commandes Basic que nous avons apprises ! Si nos questions te donnent

mal à la tête, cherche la solution aux pages 44 à 47.

## 1 Mots et espaces

```
10 CLS
20 PRINT "ROSE";"ROUGE"
30 PRINT "CHARBON";"NOIR"
40 PRINT "NEIGE";"BLANCHE"
```

Quand tu lances ce programme, les mots affichés sur l'écran sont collés les uns aux autres.

2

```
ROSE ROUGE
CHARBON NOIR
NEIGE BLANCHE
```

Modifie les lignes 20, 30 et 40 pour qu'il y ait un espace entre les mots, comme dans le dessin ci-dessus.

3

```
ROSE      ROUGE
CHARBON   NOIR
NEIGE     BLANCHE
```

Change encore ces lignes pour que les mots soient alignés en deux colonnes.

## Lignes d'étoiles

1

```
10 CLS
20 FOR K=1 TO 10
30 PRINT TAB(3);"*"
40 NEXT K
```

Essaie ce programme qui affiche une ligne d'étoiles verticale, sur le côté gauche de l'écran.

2

```
*
*
*
*
*
*
*
*
*
*
```

Change le nombre placé après TAB, pour afficher cette ligne au milieu de l'écran.

3

```
*****
```



Conseil : utilise un point-virgule pour dire à l'ordinateur de rester sur la même ligne.

Et maintenant, modifie le programme pour que l'ordinateur dessine une ligne horizontale.

## Devinette : l'invitation

```
10 CLS
20 PRINT "*****"
30 PRINT "TU ES INVITE"
40 PRINT "A MON"
50 PRINT "ANNIVERSAIRE"
60 PRINT "DATE 19 JUILLET"
70 PRINT "HEURE 15 H."
80 PRINT "LIEU CHEZ MOI"
90 PRINT "AMICALEMENT"
100 PRINT "GROS VILAIN"
110 PRINT "*****"
```



Tu peux mettre ton nom et changer les renseignements si tu le souhaites.

```
*****
      TU ES INVITE
          A
MON ANNIVERSAIRE
DATE      19 JUILLET
HEURE     15 H.
LIEU      CHEZ MOI

AMICALEMENT
GROS VILAIN

*****
```

Ce programme affiche une invitation. Mais celle-ci, tassée sur la partie gauche de l'écran, est très mal présentée. Améliore ce programme, pour que l'invitation soit au

centre de l'écran, comme sur le dessin. Tu devras remplacer les PRINT par des PRINT TAB et ajouter des lignes vides.

**4**

```
*****
 *
 *
 *
 *
 *
 *
 *
 *
 *
 *
```

Et maintenant, combine ces deux boucles pour dessiner un T sur l'écran.

**5**

```
10 CLS
20 FOR K=1 TO 10
30 PRINT TAB(K); "*"
40 NEXT K
```

Ce programme utilise PRINT TAB complété par la variable de boucle, K, pour tracer une diagonale. À chaque tour,

**6**

```
*
 *
 *
 *
 *
 *
 *
 *
 *
 *
```

l'ordinateur affiche une nouvelle étoile, décalée d'un espace vers la droite.

## Programmes de calcul

Voici deux pages de programmes de calcul.  
Essaie-les et tâche de répondre aux devinettes.

### La télévision et toi

```
10 CLS
20 PRINT "COMBIEN D'HEURES"
30 PRINT "REGARDES-TU LA TELE"
40 PRINT "PAR SEMAINE?"
50 INPUT X
60 LET Y=X*52
70 PRINT "TU REGARDES LA TELE"
80 PRINT Y;" HEURES"
90 PRINT "PAR AN!"
```

COMBIEN D'HEURES  
REGARDES-TU LA TELE  
PAR SEMAINE?  
? 6  
TU REGARDES LA TELE  
312 HEURES  
PAR AN!

Ce programme calcule le nombre d'heures que tu passes chaque année devant la télévision. L'ordinateur te demande d'abord le nombre d'heures que tu y consacres chaque semaine, puis il multiplie ta réponse par 52 (nombre de semaines que comporte une année).

### Un peu de calcul!

**1**

```
COMBIEN DE GLACES  
MANGES-TU  
PAR SEMAINE?  
? 2  
TU MANGES DONC  
104 GLACES  
PAR AN!
```

**2**

```
COMBIEN DE BONBONS  
MANGES-TU PAR JOUR?  
? 2  
TU MANGES DONC  
730 BONBONS  
PAR AN
```



Attention : tu dois  
modifier l'opération  
de la ligne 60.

Adapte le programme précédent pour calculer le nombre de glaces que tu manges en un an.

Modifions encore le programme. Mais cette fois, tu vas calculer la quantité de bonbons que tu manges en un an, à partir du nombre de bonbons que tu croques chaque jour.

## Programme de moyenne

```
10 CLS
20 PRINT "COMBIEN D'AGES"
30 PRINT "VEUX-TU DONNER?"
40 INPUT N
50 LET T=0
60 FOR K=1 TO N
70 PRINT "DONNE L'AGE NUMÉRO ";K
80 INPUT X
90 LET T=T+X
100 NEXT K
110 LET M=T/N
120 PRINT "L'AGE MOYEN EST ";M
```

Attention : l'ordinateur doit demander le temps du trajet en minutes.



Ce programme calcule la moyenne d'âge d'un groupe de personnes. Tu peux l'utiliser pour calculer celle de ta famille ou celle de tes amis. Tu rentres d'abord le nombre d'âges, puis tu tapes chaque

âge, un par un. L'ordinateur les additionne au fur et à mesure, pour trouver le total des âges (T). Ensuite, il divise T par le nombre d'âges (N) pour obtenir la moyenne.

## Devinette

```
SUR COMBIEN DE TRAJETS
PORTE LA MOYENNE?
? 5
TAPE LE TEMPS EN MINUTES
TRAJET 1
? 21
TRAJET 2
? 25
TRAJET 3
? 31
TRAJET 4
? 19
TRAJET 5
? 29
DUREE MOYENNE DU TRAJET :
25 MINUTES
```

Ce programme, tu peux facilement l'adapter pour calculer n'importe quelle moyenne. Par exemple, arrange-le pour connaître le temps que tu mets, en moyenne, pour aller à l'école.

## Opérations

**1**

```
10 CLS
20 PRINT "DONNE"
30 PRINT "DEUX NOMBRES"
40 INPUT A
50 INPUT B
60 PRINT A;" × ";B;" = ";
70 PRINT A × B
```

Avec ce programme, l'ordinateur multiplie les deux nombres que tu lui donnes. Mais il y a une erreur quelque part. Peux-tu la trouver ?

**2**

```
DONNE DEUX NOMBRES
? 4
? 2
4 + 2 = 6
4 - 2 = 2
4 × 2 = 8
```

Écris un programme qui, après t'avoir demandé deux nombres, les additionne, les soustrait et les multiplie.

**3**

```
DONNE UN NOMBRE
? 72
72 × 5 = 360
```

Écris un autre programme qui multiplie n'importe quel nombre par 5. Aide-toi de la figure ci-dessus.

## Devinette : les boucles

Essaie ces programmes pour t'habituer aux boucles.

### Devinette

```
10 CLS
20 PRINT "QU'EST-CE QUI EST ROUGE ET ROND"
30 PRINT "ET MONTE ET DESCEND?"
40 PRINT:PRINT:PRINT
50 FOR K=1 TO 2000:NEXT K
60 PRINT "UNE CERISE"
70 PRINT "DANS UN ASCENSEUR!!"
```

QU'EST-CE QUI EST  
ROUGE ET ROND  
ET MONTE ET DESCEND?

UNE CERISE  
DANS UN ASCENSEUR!!

Tu peux mettre les deux parties  
de la boucle sur une seule ligne  
à condition de les séparer par le  
signe ":".

Dans ce programme de  
devinette, la boucle de la ligne

50 oblige l'ordinateur à  
marquer une pause avant  
d'afficher le résultat.

Essaie ce programme, et  
regarde bien ce qui se passe  
sur l'écran.

## Devinette : boucles de temporisation

**1**

Voici les boucles de temporisation

```
10 CLS
20 FOR K=1 TO 100
30 PRINT "BONJOUR"
40 FOR L=1 TO 20:NEXT L
50 CLS
60 FOR L=1 TO 20:NEXT L
70 NEXT K
```

Ce programme affiche un message qui clignote  
sur l'écran. Mais ça va trop vite et on n'a pas le  
temps de lire. As-tu une solution ?

**2**

```
10 CLS
20 FOR L=1 TO 60
30 PRINT L
40 FOR J=1 TO 50:NEXT J
50 NEXT L
```

Aide-toi de ta montre ou d'une  
horloge.

Cette fois, l'ordinateur compte de 1 à 60. Quel  
nombre faut-il mettre à la ligne 40 pour que  
l'ordinateur affiche un chiffre toutes les  
secondes, comme une montre ?

## Suite de nombres

```
1  10 CLS
   20 LET R=INT(RND(1)*10+1)
   30 PRINT "VOICI TROIS NOMBRES"
   40 FOR J=2 TO 4
   50 PRINT J*R
   60 NEXT J
   70 PRINT "SI L'ON CONTINUE,"
   80 PRINT "QUEL SERA LE SUIVANT?"
   90 LET N=5*R
  100 INPUT X
  110 IF X=N THEN PRINT "BRAVO, C'EST ÇA!"
  120 IF X<>N THEN PRINT "NON, C'EST ";N
```

Jouons un peu avec les nombres. Dans ce programme, on choisit un nombre au hasard, puis on utilise une

boucle pour le multiplier trois fois. À toi de deviner le nombre suivant !

**2**

```
VOICI TROIS NOMBRES
6
9
12
SI L'ON CONTINUE,
QUEL SERA LE SUIVANT?
? 15
BRAVO, C'EST ÇA!
```



Voici ce qui apparaît sur ton écran, quand tu joues.

**3**

```
VOICI TROIS NOMBRES
24
18
12
SI L'ON CONTINUE,
QUEL SERA LE SUIVANT?
? 6
BRAVO, C'EST ÇA!
```

Conseil : utilise STEP, suivi d'un nombre négatif.

Essaie de modifier le programme pour que les chiffres aillent en diminuant. Tu devras modifier la boucle et l'opération de la ligne 90.

## Devinette

```
0
100
200
300
400
500
```

Écris le programme correspondant à ce dessin : l'ordinateur doit compter de 100 en 100, jusqu'à 500.

## Affichage

Conseil :  
mets un PRINT  
dans la boucle.



```
VOICI LE BAS  
DE L'ÉCRAN
```

Trouve le programme qui permet d'afficher simplement ce message, tout en bas de l'écran.

## Devinettes numériques

Essaie de résoudre ces devinettes, cela t'aidera à comprendre comment fonctionne la commande RND. Ensuite, complète les programmes ci-dessous avec les nombres aléatoires qui conviennent.



## Le jeu de dés

```
TON LANCER  
? 5  
MON LANCER  
3  
TU AS GAGNE
```



Maintenant, tu joues aux dés avec l'ordinateur. Lance un vrai dé et tape au clavier le chiffre que tu as tiré. Ensuite,

```
1  
LE NOMBRE TIRE AU SORT  
EST 18
```

Premier programme à inventer : l'ordinateur choisit au hasard un nombre compris entre 10 et 20 et l'affiche sur l'écran, comme ci-dessus.

```
2  
VOICI CING NOMBRES  
10  
13  
19  
17  
14
```

Second programme : l'ordinateur doit afficher 5 nombres aléatoires, toujours compris entre 10 et 20.

```
10 CLS  
20 PRINT "TON LANCER"  
30 INPUT X  
40 PRINT "MON LANCER"  
50 LET R=  
60 PRINT R  
70 IF X=R THEN PRINT "MATCH NUL"  
80 IF X>R THEN PRINT "TU AS GAGNE"  
90 IF X<R THEN PRINT "J'AI GAGNE"
```



l'ordinateur "lance" à son tour : il choisit un nombre aléatoire entre 1 et 6, et l'affiche. Attention : avant de

jouer, il faut écrire correctement l'instruction RND de la ligne 50.

## Le nombre à deviner

```
10 CLS
20 LET R=
30 PRINT "JE PENSE A UN NOMBRE"
40 PRINT "COMPRIS ENTRE"
50 PRINT "1 ET 25"
60 PRINT "TROUVE-LE"
70 INPUT G
80 IF G=R THEN GOTO 120
90 IF G<R THEN PRINT "TROP PETIT"
100 IF G>R THEN PRINT "TROP GRAND"
110 GOTO 70
120 PRINT "BRAVO, TU AS TROUVE!"
```

Dans ce jeu, l'ordinateur choisit un nombre que tu dois deviner. Quand tu proposes une solution, il te dit "TROP

PETIT" ou "TROP GRAND" pour te guider. La partie se poursuit jusqu'à ce que tu trouves. Avant de commencer,

```
JE PENSE À UN NOMBRE
COMPRIS ENTRE
1 ET 25
TROUVE-LE
? 4
TROP PETIT
? 6
TROP GRAND
? 5
BRAVO, TU AS TROUVE!
```

complète la ligne 20 pour que l'ordinateur choisisse un nombre compris entre 1 et 25.

## Devinette

**1**

```
JE PENSE À UN NOMBRE
COMPRIS ENTRE
1 ET 25
TROUVE-LE
? 8
TROP GRAND
? 7
TROP GRAND
? 3
BRAVO, TU AS TROUVE!
NOMBRE D'ESSAIS : 3
```

**2**

```
15 LET C=0
```

Attention : réfléchis bien aux numéros des lignes que tu dois ajouter.



Peux-tu compléter le programme pour que l'ordinateur compte tes essais ? Tu auras besoin d'une nouvelle variable; appelons-la C. Ajoute d'abord la ligne 15 pour donner à C la valeur 0.

Ensuite, crée une ligne pour que, à chaque tentative, l'ordinateur ajoute 1 au nombre contenu dans C. Enfin, ajoute une ligne PRINT pour afficher le nombre d'essais.

## Le jeu des erreurs

Tous les programmes de ces deux pages comportent des erreurs volontaires. Certaines provoqueront l'arrêt du programme, d'autres donneront des résultats bizarres.

Essaie de repérer ces erreurs et de les corriger pour que les programmes tournent correctement.

## Le compteur

**1**

```
10 CLS
20 PRINT "PREMIER NOMBRE?"
30 INPUT A
40 PRINT "DERNIER NOMBRE?"
50 INPUT B
60 FOR K=B TO A
70 PRINT K
80 NEXT K
```



Voici ce que donne le programme quand il fonctionne correctement.

**2**

```
PREMIER NOMBRE?
? 12
DERNIER NOMBRE?
? 15
12
13
14
15
```




Ce programme, très simple, te demande de rentrer deux nombres, A et B. Puis il fait compter l'ordinateur de A à B. Mais il y a une erreur ; peux-tu la corriger ?

## La machine à multiplier

**1**

```
10 CLS
20 PRINT "MACHINE A MULTIPLIER"
30 PRINT:PRINT
40 PRINT "QUELLE TABLE VEUX-TU?"
50 INPUT T
60 FOR J=1 TO 12
70 LET A=J*T
80 PRINT J;" x ";T;" = ";A
90 NEXT Z
```



**2**

```
MACHINE A MULTIPLIER

QUELLE TABLE VEUX-TU?
? 6
1 x 6 = 6
2 x 6 = 12
3 x 6 = 18
4 x 6 = 24
5 x 6 = 30
6 x 6 = 36
7 x 6 = 42
8 x 6 = 48
9 x 6 = 54
10 x 6 = 60
11 x 6 = 66
12 x 6 = 72
```

Si tu essaies ce programme comme cela, il ne marchera pas ! Cherche ce qui ne va pas.

Regarde bien le message d'erreur affiché par l'ordinateur : il peut t'aider.

Voici ce que donne le programme quand il fonctionne correctement.

## Jean qui pleure ou Jean qui rit

Fais tourner ce programme pour voir ce qui ne va pas. Il devrait afficher un visage souriant quand tu tapes GAI et un visage sombre quand tu tapes TRISTE. Quand tu auras repéré l'erreur, tâche de la corriger.



```

10 CLS
20 PRINT "COMMENT TE SENS-TU,"
30 PRINT "GAI OU TRISTE?"
40 INPUT A$
50 PRINT:PRINT:PRINT
60 PRINT "□////////□"
70 PRINT "!□□□□□□□!"
80 IF A$="TRISTE" THEN PRINT "!□0□□□0□!"
90 IF A$="GAI" THEN PRINT "!□-□□□-□!"
100 PRINT "!□□□,□□□!"
110 IF A$="TRISTE" THEN PRINT "!□:□□□:□!"
120 IF A$="GAI" THEN PRINT "!□□□□□□□!"
130 PRINT "!□:::□□!"
140 IF A$="TRISTE" THEN PRINT "!□□□□□□□!"
150 IF A$="GAI" THEN PRINT "!□:□□□:□!"
160 PRINT "□!□□□□□□!"
170 PRINT "□XXXXXXXX□"
180 PRINT:PRINT
190 IF A$="GAI" THEN PRINT "AIR GAI"
200 IF A$="TRISTE" THEN PRINT "AIR TRISTE"

```



Ce petit carré représente un espace. À chaque □, tape une fois sur la touche espace.



Ces lignes correspondent au dessin des yeux.



Les lignes 110 à 150 dessinent la bouche.



# Solutions

Pour trouver la réponse à une devinette, cherche le numéro de la page et le nom de la devinette. Il est possible que, pour certains jeux, tu aies trouvé des programmes légèrement différents des modèles que nous te proposons. Il faut seulement qu'ils marchent correctement.

## Page 12

### Programme du visage

```
10 CLS
20 PRINT "□□□***"
30 PRINT "□□*****"
40 PRINT "□*****"
50 PRINT "□1□□□□□1"
60 PRINT "□1□o□o□1"
70 PRINT "(□□□,□□□)"
80 PRINT "□1□□□□□1"
90 PRINT "□1□□O□□1"
100 PRINT "-----"
```



Ce programme affiche le visage dessiné page 12.

## Page 14

### Les nombres

```
LET C=11
LET G=19
LET Z=30
```

Ces instructions LET donnent aux trois variables C, G, Z les valeurs 11, 19, 30.

## Page 15

### La phrase

```
50 PRINT AS
60 PRINT BS
70 PRINT CS
```

Pour afficher les mots contenus dans les variables A\$, B\$ et C\$, il suffit d'écrire PRINT suivi du nom de la variable.

## Page 19

### L'ordinateur calculette

```
30 INPUT A
50 INPUT B
```

Voilà comment tu dois compléter ces deux lignes. L'ordinateur va ranger les nombres tapés au clavier dans les variables A et B.

### Le message

```
90 PRINT DS
110 PRINT ES
```

Pour afficher les noms stockés en D\$ ou E\$, écris PRINT suivi du nom de la variable.

## Page 21

### Programme pour un anniversaire

```
40 PRINT TAB(13);"BON ANNIVERSAIRE"
50 PRINT TAB(18);"MIMI"
60 PRINT TAB(17);"*****"
```

Pour centrer les mots, complète PRINT TAB avec les chiffres entre parenthèses. Attention : ces nombres changent suivant les ordinateurs; ceux-ci conviennent à la plupart des ordinateurs, mais sur le SPECTRUM et sur le VIC 20 il faut mettre des nombres plus petits.

## Page 27

### La boucle

```
20 FOR J=1 TO 10
30 PRINT "JOJO LE MANCHOT"
```

Complète la ligne 20 avec les nombres 1 et 10, et ajoute ton nom après le PRINT de la ligne 30.

## Page 31 Programme de loterie

```
20 LET W=INT(RND(1)*30+1)
```

Voici la commande complète. Attention : la commande RND de ton micro-ordinateur peut être différente ; reporte-toi au tableau de la page 33.

## Page 34 Mots et espaces

```
2 20 PRINT "ROSE "; "ROUGE"  
30 PRINT "CHARBON "; "NOIR"  
40 PRINT "NEIGE "; "BLANCHE"
```

Pour séparer les mots, laisse un espace après le premier mot ou avant le second. Attention : cet espace doit être à l'intérieur des guillemets.

```
3 20 PRINT "ROSE", "ROUGE"  
30 PRINT "CHARBON", "NOIR"  
40 PRINT "NEIGE", "BLANCHE"
```

Pour placer les mots en colonnes, remplace les points-virgules par des virgules. Ne laisse pas d'espace à l'intérieur des guillemets.

## Lignes d'étoiles

2

Pour que la ligne d'étoiles soit au milieu de l'écran, change le nombre placé après TAB. Il faut mettre 19 sur la plupart des ordinateurs, 15 sur le SPECTRUM, et 10 sur le VIC 20.

```
3 30 PRINT "*";
```

Pour que la ligne soit horizontale, supprime la commande TAB de la ligne 30 et ajoute un point-virgule après l'étoile.

30 est le nombre de billets achetés.



Peu importe que tu laisses l'espace avant ou après le point-virgule.



```
4 10 CLS  
20 FOR K=1 TO 10  
30 PRINT "*";  
40 NEXT K  
50 FOR K=1 TO 10  
60 PRINT TAB(5); "*"   
70 NEXT K
```

Voici le programme qui dessine un T. Il utilise les boucles des devinettes 2 et 3.

## Page 35 L'invitation

Pour centrer les lignes, remplace les instructions PRINT par PRINT TAB. Les nombres à placer entre parenthèses après TAB dépendent de ton ordinateur. À toi de les trouver !

## Page 36 Un peu de calcul!

```
1 20 PRINT "COMBIEN DE GLACES"  
30 PRINT "MANGES-TU"  
70 PRINT "TU MANGES DONC"  
80 PRINT Y; "GLACES"
```

Il faut changer les mots placés entre guillemets après PRINT.

```
2 20 PRINT "COMBIEN DE BONBONS"  
30 PRINT "MANGES-TU PAR JOUR?"  
60 LET Y=X*365  
70 PRINT "TU MANGES DONC"  
80 PRINT Y; "BONBONS"
```



Dans une année, il y a 365 jours.

Fais les changements ci-dessus. Supprime la ligne 40 en tapant 40 et en appuyant sur RETURN. À la ligne 60, X est multiplié par 365.

## Page 37

### Moyenne

```
20 PRINT "SUR COMBIEN DE TRAJETS"  
30 PRINT "PORTE LA MOYENNE?"  
45 PRINT "TAPE LE TEMPS EN MINUTES"  
70 PRINT "TRAJET ";K  
120 PRINT "DURÉE MOYENNE DU TRAJET :"  
130 PRINT A;" MINUTES"
```

Pour modifier le programme, il suffit de changer les mots des lignes PRINT. Tu dois ajouter la ligne 45 pour que le temps soit bien donné en minutes.

### Opérations

```
70 PRINT A*B
```

L'erreur se trouve à la ligne 70 : pour l'ordinateur, le signe de la multiplication, c'est \* et non ×.

```
2 10 CLS  
20 PRINT "DONNE DEUX NOMBRES"  
30 INPUT X:INPUT Y  
40 PRINT X;" + ";Y;" = ";X+Y  
50 PRINT X;" - ";Y;" = ";X-Y  
60 PRINT X;" × ";Y;" = ";X*Y
```

Avec ce programme, l'ordinateur additionne, soustrait et multiplie deux nombres, quels qu'ils soient.

```
3 10 CLS  
20 PRINT "DONNE UN NOMBRE"  
30 INPUT N  
40 PRINT N;" × 5 = ";N*5
```

L'ordinateur multiplie tout nombre tapé au clavier par 5.

## Page 38

### Boucles de temporisation

1 Il faut allonger les boucles des lignes 40 et 60 : tu dois donc mettre un nombre plus grand après TO.

2	TO7	40 FOR J=1 TO 600 :NEXT J
	VIC 20	40 FOR J=1 TO 860 :NEXT J
	Commodore 64	40 FOR J=1 TO 710 :NEXT J
	Apple	40 FOR J=1 TO 820 :NEXT J
	Spectrum	40 FOR J=1 TO 220 :NEXT J

Les ordinateurs ne fonctionnent pas tous à la même vitesse. Il faut donc adapter la longueur de la boucle à leur rapidité.

## Page 39

### Suite de nombres

```
3 40 FOR J=4 TO 2 STEP -1  
90 LET N=1*R
```

Pour compter à l'envers, il faut ajouter la commande STEP -1. La ligne 90 calcule le chiffre inférieur suivant.

### Devinette

```
10 CLS  
20 FOR J=1 TO 500 STEP 100  
30 PRINT J  
40 NEXT J
```

Pour faire compter l'ordinateur de 100 en 100, il faut introduire STEP 100 dans la boucle.

### Affichage

```
10 CLS  
20 FOR K=1 TO 18  
30 PRINT  
40 NEXT K  
50 PRINT "VOICI LE BAS"  
60 PRINT "DE L'ECRAN"
```

Sur certains ordinateurs, remplace le nombre 18 par 13.

Avec ce programme, l'ordinateur laissera 18 lignes vides pour que le message soit affiché sur les deux dernières lignes.



## Page 40

### Casse-tête numériques

```
1  10 CLS
   20 LET R=INT(RND(1)*11+10)
   30 PRINT "LE NOMBRE ALEATOIRE"
   40 PRINT "EST ";R
```

Voici le programme qui choisit un nombre compris entre 10 et 20 et l'affiche sur l'écran. Vérifie page 33 comment tu dois utiliser la commande RND sur ton ordinateur.

```
2  10 CLS
   20 PRINT "VOICI CINQ NOMBRES"
   30 FOR L=1 TO 5
   40 LET R=INT(RND(1)*11+10)
   50 PRINT R
   60 NEXT L
```

Les commandes RND et PRINT se trouvent dans la boucle : chaque fois que l'ordinateur la parcourt, il tire un nombre et l'affiche.

### Le jeu de dés

```
50 LET R=INT(RND(1)*6+1)
```

Voici comment tu dois écrire la ligne 50. Rappelle-toi que sur certains ordinateurs, il faut écrire RND au lieu de RND(1).

## Page 41

### Le nombre à deviner

```
20 LET R=INT(RND(1)*25+1)
```

Voici l'instruction complète. Rappelle-toi que sur certains ordinateurs, il faut modifier RND(1).

Entre 10 et 20,  
l'ordinateur peut choisir  
11 nombres; le plus  
petit est 10.



## Devinette

```
75 LET C=C+1
130 PRINT "TU AS FAIT ";C;" ESSAIS"
```

Il faut ajouter ces deux lignes. À la ligne 75, on augmente C de 1 à chaque tentative ; la ligne 130 affiche le nombre total d'essais.

## Page 42

### Le compteur

Dans la commande FOR... TO de la ligne 60, les variables A et B sont inversées. Il faut écrire :

```
60 FOR K=A TO B
```

### La table de multiplication

L'instruction NEXT n'est pas suivie de la bonne variable : la variable de boucle, c'est J.

```
90 NEXT J
```

## Page 43

### Jean qui pleure ou Jean qui rit

Quand tu tapes GAI, l'ordinateur dessine un visage triste, et inversement. Pour corriger, il faut donc inverser GAI et TRISTE dans les lignes comportant IF... THEN

```
80 IF AS="GAI" THEN PRINT "!□□□□□!"
90 IF AS="TRISTE" THEN PRINT "!□-□□□-□!"
110 IF AS="GAI" THEN PRINT "!□:□□□:□!"
120 IF AS="TRISTE" THEN PRINT
    "!□□□□□□!"
140 IF AS="GAI" THEN PRINT "!□□□□□□!"
150 IF AS="TRISTE" THEN PRINT "!□:□□□:□!"
```

# Les commandes Basic

Voici la liste des commandes et instructions Basic présentées dans ce

**CLS** : efface l'écran. Certains ordinateurs utilisent HOME ou CHR\$(147). (Reporte-toi au tableau de la page 33.)

**FOR... TO, NEXT** : ce sont les instructions employées pour faire une boucle. Grâce à elles, tu peux faire répéter autant de fois que tu le souhaites les instructions comprises entre FOR... TO et NEXT.

**GOTO** : ordonne à l'ordinateur de sauter à une autre ligne du programme. GOTO est suivi du numéro de la ligne où doit aller l'ordinateur.

**IF... THEN** : dit à l'ordinateur de comparer deux informations et d'agir en fonction du résultat.

**INPUT** : quand l'ordinateur rencontre cette instruction pendant l'exécution d'un programme, il s'arrête et attend que tu lui donnes une information (nombre ou chaîne). INPUT est toujours suivi d'un nom de variable.

**INT** et **RND(1)** : ce sont les commandes utilisées pour tirer un nombre au hasard. Certains ordinateurs utilisent RND ou RND(0) à la place de RND(1).

livre. Chaque commande est accompagnée d'une brève explication.

**LET** : réserve un espace en mémoire, lui donne un nom (celui de la variable) et y range une information.

**LIST** : affiche les lignes du programme sur l'écran.

**NEW** : efface le programme de la mémoire de l'ordinateur, mais généralement n'efface pas l'écran.

**PRINT** : dit à l'ordinateur d'afficher quelque chose sur l'écran. PRINT tout seul affiche une ligne vide.

**PRINT TAB** : ordonne à l'ordinateur de laisser un certain nombre d'espaces avant d'afficher quelque chose sur l'écran. Tu précises ce nombre en le mettant entre parenthèses après TAB.

**RUN** : lance l'exécution d'un programme.

**STEP** : s'utilise avec FOR... TO et NEXT dans les boucles. Permet de faire compter l'ordinateur par sauts ou à l'envers.

**STOP** : arrête l'exécution d'un programme.

ée

ne

e

r

ès

les

r

Que ne peut-on pas faire avec un ordinateur ! Calculer, bien sûr, mais aussi lui poser des questions, écrire des poèmes, jouer à quantité de jeux plus palpitants les uns que les autres, composer même de la musique...

Petits guides pratiques d'introduction à la micro-informatique, les ouvrages de cette nouvelle collection font découvrir toutes les possibilités qu'offrent les micro-ordinateurs. Ils initient au langage et au fonctionnement de l'ordinateur, apprennent à programmer et — pourquoi pas ? — à créer des programmes originaux ! La clarté du texte, la gaieté des couleurs, la drôlerie des dessins, tout est conçu dans ces livres pour faire de cette initiation un plaisir.

POUR LES PETITS



A PARTIR DE 11 ANS

#### MICRO-INFORMATIQUE

- Guide Hachette du micro-ordinateur.
- Introduction à la micro-informatique.
- Guide pratique du BASIC.
- Pratique et maîtrise du BASIC (pour écrire facilement vos programmes).
- Micro-Pratique (des program-

mes, des jeux, des montages à réaliser avec votre micro).

- Écrivez vos jeux d'aventures pour votre micro-ordinateur.
- Jeux électroniques : Battle-games.
- Jeux électroniques : Space-games.
- Jeux électroniques : fantômes et lieux hantés.

- Comment jouer avec son ordinateur et sa vidéo.

#### MATHS

- La calculatrice de poche.
- Jouer et apprendre avec sa calculatrice.

#### NOUVELLES TECHNOLOGIES

- Passeport pour les médias.
- Tout savoir sur les robots.



LELLE DE L'ÉLECTRONIQUE  
MONTBOZON

